


***Näide 5-1: taime miski parameeter sõltuvana laiuskraadist ja temperatuurist, kusjuures laius ja temperatuur korreleeruvad omavahel;**

```


data laius;
  input laius temp taim;
  cards;
  10 32 68
  20 31 38
  30 27 49
  40 26 36
  50 25 28
  60 19 36
  70 18 15
  80 14 5
  ;
options ps=35;
proc plot;
  plot laius*temp;
  plot taim*temp;
  plot taim*laius; run;

```


2) taim*temp



3) taim*laius



1) laius*temp



***5-1A: ühekaupa uurides sõltub mõlemast;**

```

proc glm;
  model taim=laius;
proc glm;
  model taim=temp;
run;

```

******* näide 5-1A output**

/...../

R-Square	Coeff Var	Root MSE	taim Mean
0.819244	25.95282	8.921283	34.37500

Source	DF	Type III SS	Mean Square	F Value	Pr > F
laius	1	2164.339286	2164.339286	27.19	0.0020

Parameter	Estimate	Standard Error	t Value	Pr > t
Intercept	66.67857143	6.95140751	9.59	<.0001
laius	-0.71785714	0.13765838	-5.21	0.0020

/...../

R-Square	Coeff Var	Root MSE	taim Mean
0.713888	32.65169	11.22402	34.37500

Source	DF	Type III SS	Mean Square	F Value	Pr > F
temp	1	1886.003472	1886.003472	14.97	0.0083

Parameter	Estimate	Standard Error	t Value	Pr > t
Intercept	-27.04166667	16.36167777	-1.65	0.1495
temp	2.55902778	0.66138162	3.87	0.0083

*** 5-1B: SS1'ga analüüsis teisenä mudelisse võetu enam ei ole oluline;**

```
proc glm;
  model taim=temp laius/ss1;
run;
proc glm;
  model taim=laius temp/ss1;
run;
```

******* näide 5-1B output: SS1 korral esimene oluline, teine mitte**

ühtepidi:

Source	DF	Type I SS	Mean Square	F Value	Pr > F
temp	1	1886.003472	1886.003472	28.15	0.0032
laius	1	420.843750	420.843750	6.28	0.0541

teistpidi:

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	2	2306.847222	1153.423611	17.21	0.0057
Error	5	335.027778	67.005556		
Corrected Total	7	2641.875000			

R-Square	Coeff Var	Root MSE	taim Mean
0.873186	23.81292	8.185692	34.37500

Source	DF	Type I SS	Mean Square	F Value	Pr > F
laius	1	2164.339286	2164.339286	32.30	0.0023
temp	1	142.507937	142.507937	2.13	0.2045

******* 5-1C: SS3'ga ei ole kumbki oluline;**

```
proc glm;
  model taim=temp laius/ss3;
run;
```

******* Näide 5-1C output:**

Parameter	Estimate	Standard Error	t Value	Pr > t
Intercept	199.0833333	91.01407477	2.19	0.0804
temp	-3.7222222	2.55233795	-1.46	0.2045
laius	-1.6750000	0.66835896	-2.51	0.0541

***** Näide 5-2. Pideva ja kategoorilise tunnuse koosmõju ja standardiseerimise vajadus *****

```
data pata;
  input trea $ wei fec;
  swei=wei-5;
  cards;
  p 3 3.5
  p 4 4.1
  p 5 4.5
  p 6 5.1
  p 7 5.5
  k 3 5.1
  k 4 6
  k 5 7.1
  k 6 8
  k 7 9.1
  ;


* pilt;
options ps=40 ls=64;
proc plot;
  plot fec*wei=trea/haxis=0 to 8 by 1
      vaxis=0 to 10 by 1;
run;

* ilma koosm;
options ps=60 ls=80;
proc glm;
  class trea;
  model fec = trea wei/ss3;
run;

proc glm;
  class trea;
  model fec = trea wei trea*wei/ss3;
run;

proc glm;
  class trea;
  model fec = trea swei
      trea*swei/ss3;
run;
```

Plot of fec*wei. Symbol is value of trea.



***** ilma koosmõjuta

Source	DF	Type III SS	Mean Square	F Value	Pr > F
trea	1	15.87600000	15.87600000	87.23	<.0001
wei	1	11.25000000	11.25000000	61.81	0.0001

***** koosmõjuga aga standardiseerimata, trea peamõju kadunud

Source	DF	Type III SS	Mean Square	F Value	Pr > F
trea	1	0.00007407	0.00007407	0.02	0.8962
wei	1	11.25000000	11.25000000	2812.50	<.0001
wei*trea	1	1.25000000	1.25000000	312.50	<.0001

***** koosmõjuga, standardiseeritud, peamõju jälle olemas

Source	DF	Type III SS	Mean Square	F Value	Pr > F
trea	1	15.876	15.87600000	3969.00	<.0001
swei	1	11.250	11.25000000	2812.50	<.0001
swei*trea	1	1.250	1.25000000	312.50	<.0001

***** Näide 5-3. Mõju suund: uurime linnu muna suuruse sõltuvust kolmest manipulatsioonist ja linnu kaalust *****

```
data solu;
  input trea $ kaal muna;
  cards;
  a 3 4
  a 4 5
  a 5 8
  a 7 7
  a 6 9
  b 2 3
  b 4 5
  b 4 4
  b 5 9
  b 6 9
  c 3 1
  c 4 4
  c 5 4
  c 5 5
  c 8 7
  ;
```

***** uurimaks, mispidi erinevused/ mõjud on, tellime lisaks solution, sest tavaline ANOVA tabel sellist ju infot ei sisalda*****

```
proc glm;
  class trea;
  model muna = trea kaal/ss3 solution;
run;
```

***** näite 5-3 tulemused, näeme, et võrreldes manipulatsiooniga c on rühmades a ja b munad suuremad ja kaalu mõju on positiivne *****

Source	DF	Type III SS	Mean Square	F Value	Pr > F
trea	2	22.16980078	11.08490039	6.34	0.0147
kaal	1	48.78048780	48.78048780	27.92	0.0003

Parameter	Estimate	Standard Error	t Value	Pr > t
Intercept	-1.897560976	1.29660118	-1.46	0.1713
trea a	2.400000000	0.83599722	2.87	0.0152
trea b	2.775609756	0.85614462	3.24	0.0078
trea c	0.000000000	.	.	0.0003
kaal	1.219512195	0.23080112	5.28	0.0003

***** Näide 5-4. Uurime rühmade vahelisi erinevusi paarikaupa. Selleks on vaja teha *multiple comparison adjustment* ja seekord seda siis Tukey meetodil *****

```
proc glm;
  class trea;
  model muna=trea kaal/ss3;
  lsmeans trea/adj=tukey tdiff;
run;
```

***** näite 5-4 tulemused *****

trea	muna LSMEAN	LSMEAN Number
a	6.27479675	1
b	6.65040650	2
c	3.87479675	3

Dependent Variable: muna

i/j	1	2	3
1		-0.43872 0.9090	2.870823 0.0462
2	0.438722 0.9090		3.241987 0.0250
3	-2.87082 0.0462	-3.24199 0.0250	

***** näide 5-4A Dunnetti test - kõigi tasemete võrdlemine ühe tasemega

<pre>proc glm; class ryhm; model vari=ryhm/ss3; lsmeans ryhm/adj=dunnett tdiff; run;</pre>	<p>The GLM Procedure</p> <p>Least Squares Means</p> <p>Adjustment for Multiple Comparisons: Dunnett</p> <p>H0:LSMean=Control</p> <table border="1"> <thead> <tr> <th>ryhm</th> <th>vari LSMEAN</th> <th>t Value</th> <th>Pr > t </th> </tr> </thead> <tbody> <tr> <td>a</td> <td>2.57142857</td> <td></td> <td></td> </tr> <tr> <td>b</td> <td>3.66666667</td> <td>1.37</td> <td>0.3174</td> </tr> <tr> <td>c</td> <td>8.00000000</td> <td>6.79</td> <td><.0001</td> </tr> </tbody> </table>	ryhm	vari LSMEAN	t Value	Pr > t	a	2.57142857			b	3.66666667	1.37	0.3174	c	8.00000000	6.79	<.0001
ryhm	vari LSMEAN	t Value	Pr > t														
a	2.57142857																
b	3.66666667	1.37	0.3174														
c	8.00000000	6.79	<.0001														

***** Näide 5-5. Jääkide arvutamine ja jääkide jaotuste uurimine eri faktorite tasemete kaupa.*******

```
proc glm;
  class trea;
  model muna=trea kaal/ss3;
  output out=jaagid r=jaak;
run;
```

```
options ps=35;
proc sort; by trea;
proc chart;
  vbar jaak;
  by trea;
run;
proc plot;
  plot jaak*kaal;
run;
```

***** Näide 5-6 - mitmene regressioon**

```
proc reg;
  model pikkus=temp valgus niisk;
run;
```

******* näide 5-6 tulemused**

Parameter Estimates

Variable	DF	Parameter Estimate	Standard Error	t Value	Pr > t
Intercept	1	-0.11505	0.19280	-0.60	0.5522
temp	1	0.59719	0.05743	10.40	<.0001
valgus	1	0.08887	0.02934	3.03	0.0032
niisk	1	0.19566	0.03413	5.73	<.0001

*** Näide 5-7 - mittelineaarsuse testimine - kas seose kirjeldamisel on abi sõltumatu muutuja ruutliikmest?;**


```
data kover;
  input x y;
  x2=x*x;
  cards;
1 7
2 2
3 8
/ /
;
options ps=35;
proc plot;
  plot y*x;
proc glm;
  model y= x x2/ ss1;
run;
```

***** näite 5-7 tulemused

The SAS System

14:17 Tuesday, March 12, 2002 106

Plot of y*x. Legend: A = 1 obs, B = 2 obs, etc.



***** näite 5-7 glm-tulemused *****

Source	DF	Type I SS	Mean Square	F Value	Pr > F
x	1	967.7218997	967.7218997	80.65	<.0001
x2	1	298.1656777	298.1656777	24.85	0.0002

Parameter	Estimate	Standard Error	t Value	Pr > t
Intercept	9.304454680	3.43910102	2.71	0.0171
x	-3.053804284	1.13780015	-2.68	0.0178
x2	0.425686914	0.08539754	4.98	0.0002

***** Näide 5-8: hierarhiline dispersioonanalüüs (nested ANOVA) - Miski sulle pikkuse sõltuvus liigist ja soost - näeme, et sugu mõjutab sulle pikkust erinevalt eri liikidel, ja soo allutamine liigile muudab tulemusi; *****

```
data pata;
  input liik $ sugu $ sulg;
  cards;
  a f 5
  a f 9
  /...../
;
proc glm;
  class liik sugu;
  model sulg = sugu liik/ss3;
run;
proc glm;
  class liik sugu;
  model sulg = liik sugu(liik)/ss3;
run;
```

***** näite 5-8 Hierarhiline anova tulemused *****

ilma nestedita;

Source	DF	Type III SS	Mean Square	F Value	Pr > F
sugu	1	0.33333333	0.33333333	0.05	0.8237
liik	1	0.33333333	0.33333333	0.05	0.8237

nestediga:

Source	DF	Type III SS	Mean Square	F Value	Pr > F
liik	1	0.33333333	0.33333333	0.16	0.6996
sugu(liik)	2	40.66666667	20.33333333	9.76	0.0071

* Näide 5-9: segatüüpi ANOVA - uuritakse, kas erinev toit mõjutab looma kaalu, loomi on aga viiest pesakonnast, pesakond siis juhuslik (random) faktor;

```
data segane;
  input pesak $ toit $ kaal;
  cards;
```

```
1 a 5
1 a 5
/...../
;
```

*** 5-9A *****

```
proc mixed;
  class pesak toit;
  model kaal=toit;
  random pesak;
```

run;

***** 5-9B on sama analüüs, aga pesakonda pole käsitletud juhusliku faktorina, st PROC MIXED võimaldab ka tavalist ANOVA't teha ja asi ei erine GLM-ga tehatvast ANOVA'st *****;

```
proc mixed;
  class pesak toit;
  model kaal=toit pesak;
```

run;

* 5-9C siin aga lubame toidu mõjul erineda ja siin tuleb juba suur vahe sisse, kas tegu on juhusliku või fikseeritud faktoriga;

```
proc mixed;
  class pesak toit;
  model kaal=toit;
  random pesak pesak*toit;
```

run;

***** 5-9D **** siin siis muutujat pesak vaadeldud fikseerituna;

```
proc mixed;
  class pesak toit;
  model kaal=toit pesak toit*pesak;
```

run;

***** näite 5-9 tulemused mixed anova - viimane ja olulisim jupp tulemustest

The Mixed Procedure
Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
toit	1	26	22.80	<.0001

***** Näide 5-10 MIXED koos mitmesuguste suvanditega,
 lihtsalt süntaksi näitena *****

```
proc mixed covtest;
  class trea pop;
  model fec = trea wei/ htype=1 outp=jaagid solution;
  random pop;
  lsmeans trea/ adjust=tukey diff;
run;
```

***** Näide 5-11 ***** korduvmõõtmiste analüüs, aeg pole
 oluline. Olukord: mitmel talvel on mõõdetud karu karva paksust,
 igal talvel paljudelt karudelt, huvitab sõltuvus talve keskmisest
 temperatuurist (üks väärtus aasta kohta)***** type=cs (compound
 symmetry) tähendab, et pole põhjust arvata, et järjestikku tehtud
 mõõtmised oleksid omavahel rohkem korreleeritud kui mistahes
 mõõtmised ***

```
data pata;
  input year temp karv;
  cards;
  2003 1 1.1
  2003 1 1.2
  /...../
  ;
proc mixed;
  class year;
  model karv = temp/solution;
  repeated/ subject = year type=cs;
run;
```

***** näite 5-11 tulemused *****

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	2.2927	0.7742	8	2.96	0.0181
temp	0.003647	0.3534	8	0.01	0.9920

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
temp	1	8	0.00	0.9920

***** Näide 5-12. Korduvmõõtmistega ANOVA, kus aeg on
 oluline, muutus ajas on just see, mis huvitab. Uurime, kuidas eri
 taimedega söötmine (trea) mõjutab linnukese parasitismiindeksit,
 seda on mõõdetud neli korda igal linnul. Type=ar(1) tähendab
 autoregressiivusust, ehk siis oletatakse, et ajas lähestikku
 tehtud mõõtmised on omavahel rohkem korreleerunud kui misiganes
 mõõtmised. *****


```

data pata;
  input nr trea $ time para;
  cards;
  1 a 1 1
  1 a 2 2
  /...../
  ;
proc mixed;
  class nr trea;
  model para = trea time trea*time;
  random nr;
  repeated/ subject = nr type=ar(1);
run;

```

***** näite 5-12 tulemused ehk siis lugu sellest, et eri taimega söötmine mõjutab küll seda, kuidas parasitismiindeks ajas muutub *****

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
trea	1	6	0.57	0.4786
time	1	22	7.12	0.0140
time*trea	1	22	7.12	0.0140

***** kogu lugu *****